Beyond Galactic Science Episode 6 - Tardis Room Schemas
Channel: Duncan Webb
Date: 23/02/2017

This video is the sixth episode of Beyond Galactic Science, a series about what to do after all the quests for Galactic Science have been completed. 

In this episode we make a Tardis Room Schema which you can download from https://www.dropbox.com/s/vha2p4p45afjyd9/TardisMod-PA-Schemas.zip?dl=0

The FTB Pack Code is THX2MYM

Music by Daria Shakhova is available at http://freemusicarchive.org/music/The_Owl/Fairy_Forest/the_owl_-_owls_secret

Other room schemas can be found at:
# https://github.com/Belgabor/TardisMod-Schemas
# https://www.dropbox.com/sh/psut2lz3d2823b5/AAA9ykYzycEZpJutC7h4aftZa?dl=0
# https://www.dropbox.com/s/09stozs1dflmewh/schemas.zip?dl=0#
# https://www.dropbox.com/s/ht1qc2ses4zljks/TARDIS%20Mod%20Schemes.zip?dl=0
# http://foxpotato.com/ (doesn't seem to work)

I'm using Avisynth and MeGUI to prepare the video for youtube and recording with Dxtory and Lagarith video codec for recording.

All the clips are wrapped with an avs script that has this (Python) template:

source = "${source}"
LoadPlugin("C:\\Program Files (x86)\\MeGUI\\tools\\ffms\\ffms2.dll")
V = FFVideoSource(source, fpsnum=30, fpsden=1, threads=1).Lanczos4Resize(1280, 720)
A1 = FFAudioSource(source, track=1).Normalize(volume=1.0, show=false)
A2 = FFAudioSource(source, track=2).Normalize(volume=1.0, show=false)
commentary = MonoToStereo(A1, A1).AmplifyDB(1.5)
audio = MixAudio(commentary, A2, 0.9, 0.1)
AudioDub(V, audio)
return last

An avs script that is common to all episodes is:

LoadPlugin("C:\Program Files (x86)\MeGUI\tools\avisynth_plugin\NicAudio.dll")

global s2 = "Until next time..."
global pngfile = "D:\dxtory\galacticscience\BeyondGalacticScienceA.png"
global endpngfile = "D:\dxtory\galacticscience\BeyondGalacticScience.png"
global musicfile = "D:\dxtory\The_Owl_-_02_-_owls_secret.mp3"

function MakeVideo(clip c, string title, int "trim_in", int "trim_out")
{
    sx = 460
    sy = 574
    font="Copasetic NF"
    font_size=28
    text_color = $d6d6c8 #light gray
    halo_color = $ecb03c #light orange
    halo_color = $df8519 #dark orange
    fadein_frames = 30
    fadeout_frames = 90
    align = 5
    title_frames = 150
    
    pngclip = ImageSource(pngfile, pixel_type="RGB32", 0, 239).Lanczos4Resize(1280, 720)
    intro = Subtitle(pngclip, title, sx, sy, font=font, size=font_size, text_color=text_color, halo_color=halo_color, lsp=60)
    endpngclip = ImageSource(endpngfile, pixel_type="RGB32", fps=30, 0, 599).Lanczos4Resize(1280, 720).FadeIn(fadein_frames)
    music = NicMPG123Source(musicfile).AudioTrim(0, 20).FadeOut(150)

    ending = AudioDub(endpngclip, music)

    trim_in = Default(trim_in, 0)
    trim_out = Default(trim_out, c.FrameCount())
    base = c.Trim(trim_in, trim_out)

    Overlay(base, pngclip, mask=pngclip.ShowAlpha().FadeOut(fadein_frames), mode="blend", opacity=0.9)
    Overlay(last, intro, mask=intro.ShowAlpha().FadeIO(fadein_frames), 0, 0) 

    FadeOut(fadeout_frames)
    Subtitle(s2, first_frame=FrameCount-title_frames-1, last_frame=FrameCount-1, size=font_size, align=align, text_color=text_color, halo_color=halo_color, lsp=60)

    last++ending
    
    ConvertToYV12(interlaced=false)
    tweak(bright=10)

    return last
}

BlankClip(width=1280, height=720)

The clips are joined with this script:
# Tardis Room Schemas
Import("D:\dxtory\galacticscience\common.avs")
Import("D:\dxtory\galacticscience\galacticscience-0013.avi.avs")
video1=last
Import("D:\dxtory\galacticscience\galacticscience-0014.avi.avs")
video2=last
Import("D:\dxtory\galacticscience\galacticscience-0015.avi.avs")
video3=last
Import("D:\dxtory\galacticscience\galacticscience-0016.avi.avs")
video4=last
TransMarbles(video1, video2, 60, 12, drop=false)
TransMarbles(last, video3, 60, 12, drop=false)
TransMarbles(last, video4, 60, 12, drop=false)
MakeVideo("Episode 6: Tardis Room Schemas")